



Weld your way.

**Carola -**

**Anforderungen**

Anforderungen an den  
Carola-Programmcode

## Inhaltsverzeichnis

<b>1</b>	<b>Einleitung</b>	<b>3</b>
<b>2</b>	<b>Gesamtanlageneffektivität (OEE)</b>	<b>3</b>
2.1	Ist-Menge und Gut-Menge bestimmen	3
2.1.1	Masterprogramm	3
2.1.2	Variablen	6
2.1.3	Einfaches Bauteilprogramm	6
2.1.4	Komplexes Bauteilprogramm	8
<b>3</b>	<b>Hinweise</b>	<b>9</b>
<b>4</b>	<b>Variablen</b>	<b>9</b>

## 1 Einleitung

Für einige Funktionen von C-Gate ist es notwendig, dass aus dem Carola-Programm Informationen an C-Gate gegeben werden. Bitte entnehmen Sie die notwendigen Blöcke und Anweisungen für die entsprechende Funktionalität den jeweiligen Kapiteln.

Falls Sie die `ReadVar` Datenstruktur bzw. den `OPCUAVAR` Befehl in einem anderen Kontext benutzen, beachten Sie, dass sich die genutzten Bereiche nicht überschneiden. Bitte beachten Sie zusätzlich, dass der genutzte Index im Carola-Programm nicht der gleiche Index auf der Weboberfläche ist. Die Werte werden durch die `ReadVar` Datenstruktur an C-Gate weitergegeben. In der Steuerung ist der erste Index eins und der letzte 100. In C-Gate ist der erste Index 0 und der letzte 99. Ein Wert, der in der Steuerung unter dem Index 50 abgelegt wird, kann in C-Gate unter dem Index 49 abgerufen werden.

Bitte beachten Sie, dass die im Folgenden erläuterten Carola-Code-Beispiele ggf. schon teilweise oder komplett im Masterprogramm hinterlegt sein können. Bitte vergleichen Sie die Abschnitte und passen Sie sie, falls notwendig, an. Eine Doppelung der Funktionen und Variablen wird vom Carola-Interpreter als Fehler erkannt.

**Hinweis:** Änderungen der QIROX `WriteVar` Variablen können den Carola Programmablauf beeinflussen.

**Hinweis:** Bitte benutzen Sie den Adobe Acrobat Reader zum öffnen dieser Datei, wenn die Sie Carola Programmcode kopieren möchten.

## 2 Gesamtanlageneffektivität (OEE)

### 2.1 Ist-Menge und Gut-Menge bestimmen

Die folgenden Programmabschnitte müssen zur Bestimmung der Ist-Menge und der Gut-Menge jeweils im Master- und Bauteilprogramm abgelegt werden. Es ist wichtig, dass die folgenden Prozeduren den Indexbereich 94 bis 100 exklusiv nutzen und nicht für andere Zwecke genutzt werden können. Der Bereich kann sich in Zukunft noch erweitern.

#### 2.1.1 Masterprogramm

Die ersten Zeilen sind nötig, damit die Variablen aus anderen Programmen importiert und geschrieben werden können. Sie sitzen direkt nach der `RESTART` Anweisung.

```
RESTART
! VARIABLENDEKLARATION
VAR SEL, ERG
STRING VAR COMPSTAT, COMPQUAL
PUBLIC STRING VAR COMPTYPE, COMPVARI, COMPSERI, COMPSTEP, COMPSEAM
```

Dieser Block definiert eine Prozedur, die an C-Gate die Information weitergibt, dass ein Bauteil beginnt. Zuvor müssen im Bauteilprogramm die unten genutzten Variablen gesetzt werden.

```
PUBLIC PROC COMPSTART
COMPSTAT := ''
OPCUAVAR(0,96, COMPSTAT, 'COMPSTATUS')
WAITM(100)
```

```

STRHND(2, COMPTYPE, COMPTYPE, ERG)
IF ERG=0 THEN COMPTYPE:='NODATA'
STRHND (0,COMPTYPE,'NODATA',ERG)
IF ERG=1 THEN BEGIN
LOPCUA1:
WRITE ('VARIABLE >COMPTYPE< NOCH NICHT BESCHRIEBEN!')
WAITM(500)
PAUSE
JUMP LOPCUA1
END

COMPSTAT := 'START'
COMPQUAL := 'IN_PROGRESS'
COMPSEAM := ''
OPCUAVAR(0,100,COMPTYPE,'COMPTYPE')
OPCUAVAR(0,99, COMPVARI,'COMPVARIANT')
OPCUAVAR(0,98, COMPSERI,'COMPSEARNO')
OPCUAVAR(0,97, COMPSTEP,'COMPSTEPNAME')
OPCUAVAR(0,96, COMPSTAT,'COMPSTATUS')
OPCUAVAR(0,95, COMPQUAL,'COMPQUALITY')
OPCUAVAR(0,94, COMPSEAM,'COMPSEAMNAME')
ENDP

```

Dieser Block definiert eine Prozedur, die an C-Gate die Information weitergibt, dass ein Bauteil beendet ist.

```

! ENDE BAUTEILEAUFZEICHNUNG
PUBLIC PROC COMPEND
COMPSEAM := ''
COMPSTAT := 'END'
OPCUAVAR(0,94, COMPSEAM,'COMPSEAMNAME')
OPCUAVAR(0,96, COMPSTAT,'COMPSTATUS')
WAITM(100)

COMPTYPE := ''
COMPVARI := ''
COMPSERI := ''
COMPSTEP := ''
COMPSTAT := ''
COMPQUAL := ''
COMPSEAM := ''
OPCUAVAR(0,100,COMPTYPE,'COMPTYPE')
OPCUAVAR(0,99, COMPVARI,'COMPVARIANT')
OPCUAVAR(0,98, COMPSERI,'COMPSEARNO')
OPCUAVAR(0,97, COMPSTEP,'COMPSTEPNAME')
OPCUAVAR(0,96, COMPSTAT,'COMPSTATUS')

```

```
OPCUAVAR(0,95, COMPQUAL, 'COMPQUALITY')
OPCUAVAR(0,94, COMPSEAM, 'COMPSEAMNAME')
WAITM(100)
ENDP
```

Dieser Block definiert eine Funktion, die an C-Gate die Information weitergibt, dass eine neue Naht startet. Der Name der Naht muss vor dem Aufruf der Prozedur in die Variable COMPSEAM geschrieben werden.

```
! START EINER NAHT
PUBLIC PROC SEAMSTRT
OPCUAVAR(0,94, COMPSEAM, 'COMPSEAMNAME')
WAITM(100)
ENDP
```

Dieser Block definiert eine Funktion, die an C-Gate die Information weitergibt, dass die aktuelle Naht beendet ist.

```
! ENDE EINER NAHT
PUBLIC PROC SEAMEND
COMPSEAM := ''
OPCUAVAR(0,94, COMPSEAM, 'COMPSEAMNAME')
WAITM(100)
ENDP
```

Dieser Block definiert eine Funktion, die auf dem PHG abfragt, ob das Bauteil in Ordnung ist oder nicht. Die Überprüfung setzt auf eine Sichtprüfung des Werkers und blockiert den weiteren Programmablauf, bis eine Eingabe gemacht wurde. Die Formulierung IST DAS BAUTEIL OK=1 ODER NICHT OK=0 ? können Sie an Ihre Anforderungen anpassen.

```
! QUALITAET BAUTEILEAUFZEICHNUNG
PUBLIC PROC COMPQMAN
WREAD ('IST DAS BAUTEIL OK=1 ODER NICHT OK=0 ?', SEL )
IF SEL=1 THEN COMPQUAL := 'OK' ELSE COMPQUAL := 'NOT_OK'
OPCUAVAR(0,95,COMPQUAL, 'COMPQUALITY')
WAITM(100)
ENDP
```

Dieser Block definiert eine Funktion, die automatisch (ohne Abfrage) die Qualität des Bauteils auf in Ordnung setzt. Diese Funktion kann genutzt werden, falls Sie die Qualität der Gesamtanlageneffektivität (OEE) nicht dynamisch erfassen wollen, aber trotzdem die gesamte Metrik korrekt berechnet haben wollen.

```
! QUALITAET BAUTEILEAUFZEICHNUNG
PUBLIC PROC COMPQAUT
COMPQUAL := 'OK'
OPCUAVAR(0,95,COMPQUAL, 'COMPQUALITY')
WAITM(100)
ENDP
```

### 2.1.2 Variablen

Die folgenden Variablen können im Programm benutzt werden.

- **COMPTYPE** : Typ des Bauteils.
- **COMPVARI** : Variante des Bauteils.
- **COMPSEAM** : Seriennummer des Bauteils. Wenn kein Wert festgelegt wird, legt C-Gate automatisch eine Seriennummer anhand eines Zeitstempels an.
- **COMPSTEP** : Name des Fertigungsschritts. Der Wert ist wichtig, wenn das Bauteil an mehreren Stationen bearbeitet wird.
- **COMPSEAM** : Name der Naht. Eine Naht kann aus mehreren Bahnsegmenten bestehen.

### 2.1.3 Einfaches Bauteilprogramm

Bitte importieren Sie die Variablen und Prozeduren aus dem Masterprogramm.

#### RESTART

```
! IMPORT DER VARIABLEN UND PROZEDUREN
EXTERNAL STRING VAR COMPTYPE,COMPVARI,COMPSEAM,COMPSTEP,COMPSEAM FROM MASTER
EXTERNAL PROC COMPSTART,COMPEND,COMPQMAN,COMPQAUT,SEAMSTRT,SEAMEND FROM MASTER
```

#### MAIN

Setzen Sie im Programmablauf mindestens die Pflichtfelder und ggf. auch die optionalen Felder. Sobald alle Felder gesetzt sind, rufen Sie die Prozedur **COMPSTART** auf. Wenn Sie das Modul C-Gate.QM erworben haben, können Sie zusätzlich Nähte definieren und das Bauteilprotokoll feiner strukturieren. Nutzen Sie dazu die Prozeduren **SEAMSTRT** und **SEAMEND**. Sobald das Bauteil fertig ist und Sie Bauteilqualität aus dem Carola-Programm setzen wollen, rufen Sie entweder **COMPQAUT** oder **COMPQMAN** auf, um die Qualitätsdaten zu übertragen. Falls Sie das Modul C-Gate.QM erworben haben, wird die Bauteilqualität automatisch mithilfe der QIROX SD-Überwachung ermittelt. Der Aufruf der Prozedur **COMPQAUT** bzw. **COMPQMAN** hat keine Auswirkung auf das Bauteilprotokoll.

Rufen Sie zum Abschluss des Bauteils die Prozedur **COMPEND** auf. Die Reihenfolge der Aufrufe ist wichtig.

Bitte beachten Sie, dass zwischen dem Aufruf von **COMPEND** und dem erneuten Start eines Bauteils mit **COMPSTART** bis zu einer Sekunde Zeit vergehen muss, da aufgrund einer Datensynchronisation in der Steuerung andernfalls das Ende-Signal verloren gehen kann. Es ist geplant, dass diese Limitierung in einer der kommenden Softwareversionen für die Robotersteuerung aufgehoben wird.

```
! SETZEN DER BAUTEILINFORMATIONEN
! PFLICHTFELD
COMPTYPE := 'SCHALTSCHRANK'

! OPTIONAL
! COMPVARI := 'XL'
```

```

! OPTIONAL
! COMPSERI := '1234567890'

! OPTIONAL
! COMPSTEP := 'BODEN'

! START DES BAUTEILS
CALL COMPSTART

! *****
! BAUTEILPROGRAMM
! *****

COMPSEAM := 'KEHLNAHT 1'
CALL SEAMSTRT

! *****
! PROGRAMM KEHLNAHT 1. ZUM BEISPIEL VERSCHIEDENE GC BEWEGUNGEN.
! *****

COMPSEAM := 'KEHLNAHT 2'
CALL SEAMSTRT

! *****
! PROGRAMM KEHLNAHT 2. ZUM BEISPIEL VERSCHIEDENE GC BEWEGUNGEN.
! *****

CALL SEAMEND

! QUALITAET DES BAUTEILS IM CAROLA PROGRAMM FESTLEGEN
! AUFRUF VOR COMPEND
CALL COMPQMAN
! ODER
! CALL COMPQAUT

! ENDE DES BAUTEILS
! AUFRUF NACH COMPQMAN ODER COMPQAUT
CALL COMPEND

! ENDE DES BAUTEILPROGRAMMS
END

```

## 2.1.4 Komplexes Bauteilprogramm

Das Folgende Carola Programm legt ein Bauteil mit zwei Fertigungsschritten an. Die Fertigungsschritte können entweder sequentiell auf einem Roboter geschweißt werden oder parallel auf mehreren. Die Voraussetzung dafür, dass die Schritte in einem Protokoll zusammengefasst werden sind:

- Typ, Variante und Seriennummer müssen auf allen Robotern gleich gesetzt sein.
- Die entsprechende Bauteilvorlage muss die korrekte minimale Fertigungsschrittzahl gesetzt haben. In diesem Fall werden zwei Fertigungsschritte geschweißt, darum muss die Zahl mindestens auf den Wert zwei gesetzt sein.

```
RESTART
EXTERNAL STRING VAR COMPTYPE,COMPVARI,COMPSEAM FROM MASTER
EXTERNAL PROC COMPSTART,COMPEND,COMPQMAN,COMPQAUT,SEAMSTRT,SEAMEND FROM MASTER
MAIN

! *****
! FERTIGUNGSSCHRITT 1
! *****

COMPTYPE := 'KESSEL'
COMPVARI := '80KW'
COMPSEAM := '0042012342'
COMPSTEP := 'BODEN'

CALL COMPSTART

COMPSEAM := 'KEHLNAHT 1'
CALL SEAMSTRT
! PROGRAMM KEHLNAHT 1. ZUM BEISPIEL VERSCHIEDENE GC BEWEGUNGEN.

COMPSEAM := 'KEHLNAHT 2'
CALL SEAMSTRT
! PROGRAMM KEHLNAHT 2. ZUM BEISPIEL VERSCHIEDENE GC BEWEGUNGEN.

CALL SEAMEND
CALL COMPQAUT
CALL COMPEND

! *****
! FERTIGUNGSSCHRITT 2
! *****

COMPTYPE := 'KESSEL'
COMPVARI := '80KW'
COMPSEAM := '0042012342'
```



```

COMPSTEP := 'DECKEL'

CALL COMPSTART

COMPSEAM := 'KEHLNAHT 3'
CALL SEAMSTRT
! PROGRAMM KEHLNAHT 3. ZUM BEISPIEL VERSCHIEDENE GC BEWEGUNGEN.

COMPSEAM := 'KEHLNAHT 4'
CALL SEAMSTRT
! PROGRAMM KEHLNAHT 4. ZUM BEISPIEL VERSCHIEDENE GC BEWEGUNGEN.

CALL SEAMEND
CALL COMPQAUT
CALL COMPEND

END

```

### 3 Hinweise

Falls Sie PDM- bzw. UMS-Kunde sind oder waren und eine der beiden Optionen in der Konfiguration der Steuerung aktiviert ist, muss der folgende Hinweis dringend beachtet werden! Falls eine der beiden Optionen aktiviert ist, aber die Robotersteuerung nicht mit der entsprechenden PC-Anwendung verbunden ist, kommt es zu Lücken im Datenstrom, die bis zu 5 Sekunden lang sein können und das Bauteilprotokoll signifikant verfälschen. Bitte verbinden Sie entweder die Robotersteuerung mit der PC-Anwendung oder lassen Sie die entsprechende Option aus der Konfiguration entfernen.

Falls Sie mit mehreren Robotern gleichzeitig am selben Bauteil arbeiten und alle Daten in einem Bauteilprotokoll im Rahmen des Moduls C-Gate.QM zusammengeführt haben möchten, kontaktieren Sie bitte den CLOOS Service für weiterführende Informationen. Andernfalls lassen Sie bitte das Feld `COMPSEAM` leer, da es sonst zu unerwünschten Nebeneffekten bei der Berechnung der Gesamtanlageneffektivität (OEE) kommen kann.

### 4 Variablen

Sie haben die Möglichkeit aus dem Carola Programm Variablen an das Bauteilprotokoll zu hängen. Dazu muss man das Kommando `OPCUAVAR` benutzen (siehe Beispiel). Wenn der Name (letzter Parameter) das Präfix `'COMPVAR_'` hat, wird die Variable an das Bauteil gehängt. Wenn der Name das Präfix `'STEPVAR_'` hat, wird die Variable an den Bauteilschritt gehängt. Wenn der Name das Präfix `'SEAMVAR_'` hat, wird die Variable an die aktuelle Naht gehängt. Zu jedem gewählten Namen kann nur ein Wert abgelegt werden. Der letzte Wert, der zuletzt mit dem jeweiligen Namen gesetzt wurde, wird im Protokoll abgelegt. Das Präfix vom Namen wird nicht abgespeichert. Der Wert muss vor Abschluss der Naht, der Schritt bzw. des Bauteils zurückgesetzt werden. Falls das nicht gemacht wird, wird der Wert auch an das folgenden Element gegangen. Es können Ganzzahlen, Kommazahlen und Texte abgespeichert werden.

```

CALL SEAMSTRT
OPCUAVAR(0,1, 'EXAMPLE-VALUE', 'COMPVAR_SEAM_MEASURE_A')

```

```
OPCUAVAR(0,1, '', '')  
CALL SEAMEND
```